



Excel voor beginners (3)

Handleiding van Helpmij.nl

Auteur: Corverm

Mei 2008

“

Dé grootste en gratis computerhelpdesk van Nederland

”

Excel voor beginners deel 3.

Vorige delen:

[Deel 1](#)

[Deel 2](#)

In het vorige deel hebben we beloofd om iets te gaan maken i.v.m. het EK voetbal, en hoe dan ook belofte maakt schuld. Eenmaal begonnen aan dit deel heb ik me pas gerealiseerd dat de stap die we maken ten opzichte van de vorige delen best wel groot is. In ieder geval zal ik m'n best doen om zo duidelijk mogelijk te zijn en met een zo helder mogelijke uitleg verder te gaan met waar we de vorige keer gebleven zijn.

Eerst weer wat theorie.

Is het je opgevallen dat tijdens het typen van een formule de formuleopbouw in de betreffende cel te zien is? Ook in de formulebalk kun je de voortgang zien van de formuleopbouw, alleen moet je dan wel even in de formulebalk klikken, anders zie je alleen de getypte tekst. Als je =ALS(hebt getypt verschijnt de volgorde van de formuleopbouw in beeld, het vet gedrukte deel is aan de "beurt". Heb je de logische-test (de voorwaarde waaraan voldaan moet worden) getypt dan volg een ; waarna het volgende deel van de formule vet in beeld komt. Uiteindelijk sluit je de formule af met een haakje. Zo meteen in de praktijk meer daarover.



Excel heeft vijf Logische functies in huis, ALS(), EN(), NIET(), OF(), ONWAAR() en WAAR(). Deze functies kun je onderling combineren (nesten), waarbij opgemerkt dat je in een formule zeven maal kunt nesten.

In dit deel komen de nodige formules voorbij evenals de functie Voorwaardelijke opmaak. Zo goed als dit hele model bestaat uit voorwaarden waarvan hierboven een voorbeeld te zien is. Hier gaat het om een Logische test. Uiteraard komen we daar straks op terug.


Wanneer je tekst wilt weergeven middels een formule moet je de tekst altijd tussen aanhalingstekens plaatsen, aan het eind van dit deel zal daar een voorbeeld van te zien zijn. Gaat het erom een getal weer te geven middels een formule dan moet je dat zonder aanhalingstekens opnemen in de formule.

Ook een aantal statistische functies komt aan de beurt, zoals de functie AANTAL.ALS(). Deze functie "kijkt" hoe vaak een bepaalde waarde voorkomt in een bepaald bereik.

De functie SOM.ALS() telt de waarden op die aan een bepaalde voorwaarde voldoen.

Aan het werk dan maar.


Maak onderstaande model maar na op de uitslagen na, die komen later aan bod. Geef de cellen waar Nederland in staat de kleur oranje door ze te selecteren (met ingedrukte Ctrl-toets) en bij Opvulkleur

de juiste kleur te kiezen. 

	A	B	C	D	E	F	G	H	I	J	K	L
1												
2		Datum	Tijd		Thuis	Uit	Uitslagen					
3							thuis	uit		gew.	verl.	
4		9-6-2008	18:00	C	Roemenië	Frankrijk	3	2	5	Roemenië	Frankrijk	
5		9-6-2008	20:45	C	Nederland	Italië	3	3	6	Gelijk	Gelijk	1
6		13-6-2008	18:00	C	Italië	Roemenië	1	1	2	Gelijk	Gelijk	1
7		13-6-2008	20:45	C	Nederland	Frankrijk	1	2	3	Frankrijk	Nederland	
8		17-6-2008	20:45	C	Frankrijk	Italië	2	1	3	Frankrijk	Italië	
9		17-6-2008	20:45	C	Nederland	Roemenië	1	0	1	Nederland	Roemenië	


Voor de opvulkleur in de kolommen J en K gaan we een heel ander verhaal vertellen. Daarvoor gebruiken we de functie Voorwaardelijk opmaak, we weten immers van te voren niet in welke kolom Nederland terecht komt.

Selecteer de cellen J4:K9, ga naar het menu Invoegen en kies daar voor Voorwaardelijke opmaak.



In het scherm Voorwaardelijke opmaak staat standaard “Celwaarde is” geselecteerd. Klik op het pijltje in het vakje waar “gelijk aan” staat en kies deze optie. Typ vervolgens Nederland (zonder het = teken of de aanhalingstekens) en klik op de knop Opmaak. In het scherm dat dan verschijnt kies je voor de tab Patronen en vervolgens voor oranje door op het betreffende vakje te klikken. Klik tweemaal op OK om de vensters te sluiten. Elk van deze cellen zal oranje kleuren als “Nederland” daarin verschijnt. Maar wat nu? Als Nederland gelijk speelt zien we geen oranje cel(len) terug. Vandaar dat we een tweede voorwaarde toe gaan voegen.

Let op! Klik J4 aan, ga weer het menu Invoegen > Voorwaardelijke opmaak en klik in het scherm “Voorwaardelijke opmaak” op de knop toevoegen.



Kies nu voor “Formule is” en neem de formule over zoals je die ziet staan in de afbeelding, waarbij opgemerkt dat je nu wel alle tekens moet typen. Druk weer op de knop Opmaak en kies weer voor de opmaak zoals beschreven. Twee maal op OK klikken en ook dit is gefikst. Bijna dan, want we moeten

de opmaak nog kopiëren naar de andere cellen.  Klik op de kwast Opmaak (je hebt nog steeds J4 geselecteerd staan) en klik op J5, houdt de muisknop ingedrukt en sleep de cursor (in de vorm van een kwast) tot aan J9. Nu hebben ook deze cellen de goede (voorwaardelijke) opmaak. Voor de cellen in kolom K geldt hetzelfde verhaal.

Bewust is hier gekozen voor een wat omslachtige manier van werken, het kan wat directer maar nu weet je hoe het vanaf de bodem in z'n werk gaat.

Je ziet dat alle uitslagen zijn ingevuld, voor het Nederlands elftal niet te hopen dat het zo afloopt als in dit voorbeeld, dus kun je deze uitslagen maar beter niet overnemen. Wat je wel moet doen is een kleine formule plaatsen in I4 (voor de duidelijkheid: ie4). In I4 komt: =ALS(OF(G4="";H4="");"";G4+H4), (zonder komma) deze formule telt het aantal doelpunten op die in de wedstrijd gevallen zijn. Je zou in dit geval kunnen volstaan met: =G4+H4, 't is maar dat het aantal doelpunten van de wedstrijd bij elkaar opgeteld worden. 't Is ook een mooie gelegenheid om te laten zien hoe je twee logische functies kunt gebruiken in een formule.

```
=ALS(OF(G4="";H4="");"";G4+H4)
```

Door logische functies te nesten kun je meerdere voorwaarden (7X) gebruiken.

Het eerste deel van de formule kijkt of de cellen G4 en H4 leeg zijn, (="" ,je weet nog wel dat het hier 2X een dubbel aanhalingteken betreft) is dat het geval dan moet I4 ook leeg blijven. Pas als er G4 en H4 iets ingevuld is komt de som in I4 te staan. Let op de haakjes, om de OF() functie af te sluiten typ je een) haakje. Door een ; te typen ga je verder met de ALS() functie. Goed, I5 is nog geselecteerd, druk op de Shift-toets en klik op I9, druk vervolgens op de toetsen Ctrl-D om de formule naar beneden te kopiëren. De truc met de vulgreep ken je al.

Thuis	Uit	Uitslagen		3	gew.	verl.	
		thuis	uit				
Roemenië	Frankrijk	1	2	3	Frankrijk	Roemenië	
Nederland	Italië	3	3	6	Gelijk	Gelijk	1

Als het even kan willen we ook wel weten wie er gewonnen, verloren of gelijk gespeeld heeft. Zet in J4 de formule: =ALS(G4="";"";ALS(G4=H4;"Gelijk";ALS(G4>H4;E4;F4))), een hele mond vol zo lijkt het. Eerst kijken we of G4 leeg is, in dat geval moet J4 ook leeg blijven. Daarna kijken we (nou ja, de formule natuurlijk) of de uitslag in G4 en H4 gelijk is, in dat geval moet er in J4 het woord Gelijk verschijnen. Is de uitslag in G4 meer dan (>) H4 dan komt in J4 de tekst uit E4 te staan en anders de tekst uit F4. Duidelijk zo?

In K4 zetten we de formule: =ALS(H4="";"";ALS(G4=H4;"Gelijk";ALS(H4>G4;E4;F4))). We selecteren J4 en K4 en "trekken" de formules met de vulgreep naar beneden tot en met J9 en K9.

We hebben nu driemaal de ALS() functie genest hetgeen betekent dat we de formule met drie haakjes moeten afsluiten. Het is altijd zo dat er, hoe dan ook, evenveel haakjes dicht als haakjes open in de formule gebruikt moeten worden.

We nemen de cellen L4:L9 nog onder de loep.

```
=ALS(EN(J4="gelijk";K4="gelijk");1;"")
```

Ditmaal is =ALS(EN(aan de beurt. (Hebben we trouwens al verteld dat dit soort formules **altijd** beginnen met de functie ALS(). Nee? Bij deze dan.) Ditmaal doet het er wel degelijk toe of aan beide voorwaarden wordt voldaan. Want pas als er in de cellen J4 en K4 het woord gelijk staat moet er in L4 een 1 komen te staan en anders moet L4 leeg blijven.

Om te zien hoe de vlag er verder bij hangt maken we het onderstaande model na, uitgezonderd de getallen. Want die komen vanzelf voor de dag als er, uitslagen worden ingevuld. Maar eerst gaan we natuurlijk de nodige formules invoeren.

	M	N	O	P	Q	R	S	T	U	V	W
1											
2	GROEP C							Doelpunten	Doelpunten		
3		gespeeld	gew.	verl.	gelijk		punten	totaal	voor	tegen	saldo
4	Nederland	3	1	1	1		4	10	5	5	0
5	Italië	3	0	1	2		2	11	5	6	-1
6	Roemenië	3	1	1	1		4	8	4	4	0
7	Frankrijk	3	2	1	0		6	11	6	5	1
8											

In N4 zetten we de formule: =SOM(O4:P4;Q4), hoewel de cellen O4, P4 en Q4 dan nog leeg zijn zal Excel een nul laten zien omdat Excel in dit geval een lege cel ook voor nul aanziet.

In O4 komt de formule: =AANTAL.ALS(J\$4:J\$9;M4).


```
=AANTAL.ALS(J$4:J$9;M4)
```

AANTAL.ALS(bereik; criterium)

Als eerste geef je het bereik op waar Excel moet zoeken, als tweede het criterium waarop Excel moet zoeken. Concreet gezegd moet er in het bereik J\$4:J\$9 gekeken worden hoe vaak daar de waarde uit M4 (Nederland) in voorkomt. De dollartekens voor de celverwijzing zijn nodig om tijdens het kopiëren van de formule het juiste bereik vast te houden. (Zie deel twee voor relatieve en absolute formules) In P4 zetten we: =AANTAL.ALS(K\$4:K\$9;M4) zodat we ook kunnen zien hoe vaak Nederland heeft verloren. Je weet onderhand wel dat je de formules tot in rij negen moet kopiëren.

Dan is Q4 aan de beurt, we kijken daar hoe vaak er gelijk gespeeld is met:
 =SOM.ALS(E\$4:E\$9;M4;L\$4:L\$9)+SOM.ALS(F\$4:F\$9;M4;L\$4:L\$9)

```
=SOM.ALS(E$4:E$9;M4;L$4:L$9)+SO
```

SOM.ALS(bereik; criterium; [optelbereik])

Ook hier gaan we kijken of in het opgegeven bereik het opgegeven criterium voorkomt om, als dat zo is, in het optelbereik de som te maken. Concreet: als in E\$4:E\$9 Nederland (M4) voorkomt dan tel je het aantal enen op die in het bereik L\$4:L\$9 achter Nederland staan. Om alles compleet te maken moeten we ook in het bereik F\$4:F\$9 kijken of ook daar het gezochte land in voorkomt, om middels het + teken de twee uitkomsten bij elkaar op te tellen.

Om het aantal bij elkaar gespeelde punten te bepalen komt in S4 de formule: =(O4*3)+(Q4*1)

Dan hebben we het totaal aantal doelpunten nodig om te kunnen bepalen hoeveel er voor en tegen gescoord is met in T4: =SOM.ALS(E\$4:E\$9;M4;I\$4:I\$9)+SOM.ALS(F\$4:F\$9;M4;I\$4:I\$9)

Voor het aantal doelpunten voor gebruiken we in U4:
 =SOM.ALS(E\$4:E\$9;M4;G\$4:G\$9)+SOM.ALS(F\$4:F\$9;M4;H\$4:H\$9)

Om voor het aantal tegendoelpunten in V4 =T4-U4 te zetten.
 Het saldo wordt berekend door =U4-V4 in W4 te zetten.
 Niet vergeten om de formules te kopiëren.


	A	B	C	D	E	F	G	H	M	N	O	P	Q	R	S	U	V	W
1																		
2		Datum	Tijd		Thuis	Uit	Uitslagen		GROEP C							Doelpunten		
3							thuis	uit		gespeeld	gew.	verl.	gelijk		punten	voor	tegen	saldo
4		9-6-2008	18:00	C	Roemenië	Frankrijk			Nederland	0	0	0	0		0	0	0	0
5		9-6-2008	20:45	C	Nederland	Italië			Italië	0	0	0	0		0	0	0	0
6		13-6-2008	18:00	C	Italië	Roemenië			Roemenië	0	0	0	0		0	0	0	0
7		13-6-2008	20:45	C	Nederland	Frankrijk			Frankrijk	0	0	0	0		0	0	0	0
8		17-6-2008	20:45	C	Frankrijk	Italië												
9		17-6-2008	20:45	C	Nederland	Roemenië												
10																		

Een aantal kolommen hoeven we niet meer te zien en dus gaan we die verbergen. Selecteer de kolommen I t/m L en kolom T door op de kolomletter te klikken met de Ctrl-toets ingedrukt, een maal geselecteerd klik je met rechts op een van de kolommen en kies je voor Verbergen. Het geheel ziet er dan zo uit. Vind je het netter staan om tussen de uitslagen en de groep een vrije ruimte te zien, klik dan met rechts op kolomletter M en kies voor Invvoegen. De formules passen zich vanzelf aan de nieuwe situatie aan.

GROEP C	gespeeld	gew.	verl.	gelijk	punten	Doelpunten			rang
						voor	tegen	saldo	
Nederland	3	1	1	1	4	5	5	0	2
Italië	3	0	1	2	2	5	6	-1	4
Roemenië	3	1	1	1	4	4	4	0	2
Frankrijk	3	2	1	0	6	6	5	1	1

Tenslotte zouden we in het groepsoverzicht ook de juiste rangschikking wel willen zien en dus gaan we daar wat aan doen. Geef eerst ook de cellen M4:M7 de voorwaardelijke opmaak voor Nederland.

In X4 komt de formule: =RANG(S4;S\$4:S\$7;0) die je kopieert tot in X7.



Aangezien we van het puntenaantal de Rang gaan bepalen nemen we voor **getal** S4, **verw** staat voor het bereik waaruit de Rang moet worden bepaald (S\$4:S\$7). Dan **volgorde** dat tussen vierkante haken staat, wat in dit geval betekent dat het een optioneel onderdeel van de formule betreft. Laat je de "volgorde" weg dus =RANG(S4;S\$4:S\$7) dan zal Excel altijd rangschikken op de hoogste volgorde, dus in het voorbeeld heeft Italië 7 punten en krijgt de rang 1 mee. Hier, in het voorbeeld, is wel **volgorde** gebruikt en geeft hetzelfde resultaat omdat er een nul als volgorde is opgegeven. Zou daar een 1 staan dan zou Italië rang 4 hebben, dus niet oplopend maar aflopend gerangschikt. Zolang we nog geen uitslagen hebben ingevoerd staan alle gegevens in het groepsoverzicht op nul, hetgeen betekent dat de rangschikking in alle gevallen op 1 komt te staan. We kunnen dit ondervangen door de volgende formule te gebruiken: =ALS(N4=0;"nog geen rang";RANG(T4;T\$4:T\$7)). In kolom N komen het aantal gespeelde wedstrijden te staan, en dus in tegenstelling tot de andere kolommen (hangt af van winst, verlies of gelijkspel) altijd optellen. Onnodig te zeggen dat ook deze formule naar beneden moet worden gekopieerd. Je kunt er ook voor kiezen om tekst weer te geven zolang er niets te sorteren valt, dat doe je zo: =ALS(N4=0;"nog geen rang";RANG(S4;S\$4:S\$7)), m.a.w. is de waarde in N4 0 geef dan de tekst weer. In dit geval komt er dan: nog geen rang te staan. Nu we de rang bepaald hebben kunnen we gaan sorteren. Selecteer M4:X7, ga in het menu Data naar Sorteren en kies bij "Sorteren op" voor Kolom X door op het pijltje te klikken.



Let even op dat de optie Oplopend is geselecteerd evenals de optie Geen veldnamenrij. Alhoewel Excel dat laatste ook wel in de gaten heeft kan het geen kwaad om dat toch even te controleren.

GROEP C	gespeeld	gew.	verl.	gelijk	punten	Doelpunten			rang
						voor	tegen	saldo	
Frankrijk	3	2	1	0	6	7	5	2	1
Roemenië	3	1	0	2	5	6	5	1	2
Nederland	3	0	1	2	2	6	8	-2	3
Italië	3	0	1	2	2	5	6	-1	3

Het resultaat van de sorteeractie.

Rangschikken op alleen het aantal punten is een hachelijke zaak. Wat nu als er twee (of misschien wel meer landen) hetzelfde aantal punten behalen? Hier ligt een mooie uitdaging voor jou om dat

vraagstuk op te lossen. Je kunt dan denken aan een som van punten en saldo, maak daar dan een aparte kolom voor en laat Excel daar dan op rangschikken. Nou ja, laat je creativiteit maar spreken.

In principe ben je nu in staat om een heel speelschema op te bouwen aan de hand van het bovenstaande. Houdt er rekening mee dat je de bereiken aanpast aan de nieuwe situatie.

Tenslotte nog een opmerking over de gebruikte formules.

In de formules hebben we steeds gebruik gemaakt van celverwijzingen, terwijl het ook mogelijk is om met bereiknamen te werken. Goed, het is er om te doen dat je met Excel kan leren werken dus moeten alle opties besproken worden. En dan begin je bij het begin. Even hebben we er aan gedacht om ook hier het Naam geven te bespreken maar dat is misschien wel teveel van het goede in een keer. Vandaar dat we in de volgende aflevering daar op terug komen.

Helpmij.nl